It's Time to Set Your RPG Free

Susan Gantner

Susan.Gantner @ Partner400.com
www.Partner400.com
www.SystemiDeveloper.com

RPG & DB2 Summit
A System i Developer Event

Notes

Expressions and limited free-form calculations have been part of our RPG IV coding lives since the inception of the language. Most of us jumped at the chance to write our logic in free format introduced in V5.1. Then came V7.1 and with it the ability for programs to be coded completely in free-format - even the declarations. Now, we're not even limited to columns 8 to 80.

Perhaps you've heard about all these features - you may tried your hand at writing some free-form declarations and/or logic. But many RPGers still stumble over "how do I ... ?" issues and wonder if it is really worth the effort.

If this describes you then come to this session where Susan will do her best to clear up any misconceptions you may have and show you why and how to take this next step. We'll even see how RDi (the only editor from IBM that supports free-format RPG) can help ease the transition.

In this session, Susan will explain:

- Why free format RPG is important
- What are the rules?
- Just how free can you go?
- Are there any gotchas?
- How can RDi help?
- Are there tools available to convert fixed format code easily?

Susan Gantner is co-founder of Partner400, a firm specializing in customized on-site education and mentoring services for IBM i developers.

She is also a partner in System i Developer, a consortium of IBM i educators who host events such as the RPG & DB2 Summit.

Susan and her partner, Jon Paris, author regular technical articles for IBM Systems Magazine, IBM i edition, and the companion electronic newsletter, IBM i EXTRA. You may view articles in current and past issues and/or subscribe to the free newsletter or the magazine at: www.ibmsystemsmag.com/IBMi View Jon & Susan's weekly blog at ibmsystemsmag.com/Blogs/iDevelop/

This presentation may contain small code examples that are furnished as simple examples to provide an illustration. These examples have not been thoroughly tested under all conditions. We therefore, cannot guarantee or imply reliability, serviceability, or function of these programs.

All code examples contained herein are provided to you "as is". THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE EXPRESSLY DISCLAIMED.

Agenda

Why Free-form?

How Free do you want to go?
- Free-form logic
- Free-form declarations
- Freedom from column 8 - 80 restriction

Rules & syntax for all 3 options
- And Gotchas

Tools to help
- How RDi helps
- Tools to convert fixed to free



Notes

Free-form calculations in RPG IV first became available in V5R1 but it was not until V7.1 that data declarations (the former D-specs) and file declarations (F-specs) became available in free-form. Then in V7.2 (also supported in V7.1) we have the ability to do all the free-form coding without the 8-80 column restrictions.

Why Free-form ?

Partner400

1. Indented code is more understandable

- Match up nested If ... EndIf, DoW ... EndDo, etc.
- Nesting of levels in data structures

2. More efficient use of space in source code

- No more wasted space between C and Op Code column
- Complex expressions take fewer lines

3. More room for larger names

- More descriptive and/or qualified data names easier to code and use

4. New features in Free-form only

- Freed from the limitations of 2 Factors and a Result field or D spec columns

5. Much easier to read (and write!) declarations (esp. for files)

- Keywords with intelligent default values

6. Attraction of new RPG programmers

- Free-form looks and feels more like other modern languages
- Do you want your legacy to be discarded?

Notes

1) When writing free format logic, I can indent the statements to show the structure of the code. The code is more easily understood, especially in cases involving more complicated logic blocks, such as nested If statements, Select/When statements or Monitor blocks. The structure of data declarations - especially data structures - can also be made more obvious in free format. The easier the code is to understand, the faster and less error prone it will be to maintain.

2) I find that I seldom use Factor 1 in RPG IV. At the same time, because I am no longer limited to simple arithmetic operations, many expressions - especially those involving built-in functions - get quite lengthy, often requiring multiple lines to complete. Therefore with fixed format logic, the source edit screen typically has a huge amount of unused space on the left (i.e., where Factor 1 and the outdated left-hand conditioning and cycle control indicators would go) while the right side is complicated with many multi-line statements. Using free format logic, I have much more space for my expressions. My edit screens are "cleaner" and I can see more logic at a time because there are fewer multi-line statements.

3) There are no real length limitations for names when you can code declarations and logic in free format. I don't often set out to create names of 20+ characters. But it's nice not to have to think of an acceptable abbreviation for "ProcessCustomer" or "ErrorMsgDisplay", which are just slightly too long to fit into the old fixed format columns. Qualified data names allow powerful features such as data structure arrays and nested data structures, but could never have fitted into a 14 character Factor column.

4) Some new language functions are only available in free format logic. This is because a column-oriented language such as traditional RPG puts severe limitations on some kinds of new features that require significant space in the statement for implementation.

5) The V7 free format F and D spec replacements make data and file declarations far easier to read. No need to remember which characters go into which columns in the F spec. No more copying an F spec from another program because you can't be bothered to figure out which columns were which! An excellent set of default values means you rarely need to specify all the keywords.

6) Last, but not least, free format coding brings RPG more in line with other modern programming languages, all of which use free format logic. This is important for attracting new developers coming into the marketplace. Traditional fixed format RPG less attractive and gives RPG the undeserved appearance of an old-fashioned language not up to modern application tasks. RPG is a powerful and flexible language that many young developers come to prefer over other more popular language options for business applications. But they must first be attracted to learn the language.

Old and New - Free Form Logic

```
C                   READ(E)   TransFile
C                   DOW       Not %EOF(TransFile)
C                   IF        %Error
C                   Eval      Msg = 'Read Failure'
C                   LEAVE
C                   ELSE
C     CustNo        CHAIN(N)  CustMast      CustData
C                   Eval      CustName = %Xlate(UpperCase:LowerCase
C                                        : CustName)
C                   EXSR      CalcDividend
C                   READ(E)   TransFile
C                   ENDIF
C                   ENDDO
```

```
READ(E) TransFile;
 DOW not %EOF(TransFile);
    IF %Error;
       Eval Msg = 'Read Failure';
       LEAVE;
    ELSE;
       CHAIN(N) CustNo CustMast CustData;
       Eval CustName = %Xlate(UpperCase : LowerCase : CustName);
       EXSR CalcDividend;
       READ(E) TransFile;
    ENDIF;
 ENDDO;
```

Notes

This small example of a comparison of fixed format RPG IV and Free-form format logic illustrates some of the advantages described on the earlier chart.

Old and New - Free Form Declarations

```
FCustMast  UF A E           K Disk
FReport    O    E             Printer OflInd(*IN96)
FScreen    CF   E             Workstn

D MyDS             DS
D  Address                       32A
D    Street                      15A   Overlay(Address)
D    City                        10A   Overlay(Address: *Next)
D    State                        2A   Overlay(Address: *Next)
D    Zip                          5A   Overlay(Address: *Next)
D  LstOrdDate                     D    DatFmt(*USA)
D  VaryingField                  25A   Varying
```

```
Dcl-F CustMast Usage(*Delete:*Output) Keyed;
Dcl-F Report   Printer OflInd(*IN96);
Dcl-F Screen   Workstn;

Dcl-DS MyDs;
 Address        Char(32);
    Street       Char(15) Overlay(Address);
    City         Char(10) Overlay(Address:*Next);
    State        Char(2)  Overlay(Address:*Next);
    Zip          Char(5)  Overlay(Address:*Next);
 LstOrdDate   Date(*USA);
 VaryingField Varchar(25);
End-DS;
```

Notes

This small example of a comparison of fixed format RPG IV and Free-form format declarations illustrates some of the advantages described on the earlier chart.

How Free is Free? A History Lesson

Partner400

RPG IV was introduced in V3R1 (1994)

- Still mostly columnar except for extended Factor 2 in logic
  - Extended Factor 2 and built-in functions allowed for complex expressions
  - Extended columns allowed for somewhat longer names

Completely free format logic arrived in V5R1 (2001)

- Required /Free directive before all uses of free format
  - And /End-Free to insert fixed-format logic or declarations - e.g., P specs
- Allowed for indentation to illustrate nested logic
- Enabled some new features

Free format declarations finally arrived in V7.1 (2012)

- Free format replacements for H, F, D and P specs
  - Indentation for structures and intelligent defaults ease coding
- Still limited to coding between columns 8 & 80

Note: SEU editor does not support V7+ features. Use RDi!

V7.2 (also supported in V7.1) did away with 80 column restrictions (2015)

- First statement must be **Free
- No fixed-format code allowed at all in the same source member
- But can be added via /Copy

RPG IV dipped its toe into free-format logic in the beginning with the semi-free-format expressions allowed with the extended factor. You may not know it, but fully free-form was part of the original design for RPG IV. But for a lot of reasons it didn't happen until much later in RPG IV's life. Jon has written a series of blog posts on his perspective of this story and you can find links to all the articles in the series here: systemideveloper.com/blogs/part-6-of-the-history-of-rpg-published/

True free-format arrived in V5R1 - but only for logic statements. The /Free directive was required before any free format logic could be coded and the /End-free directive was required before anything that wasn't free-format was coded. This could include either fixed-format logic (perhaps because of using an op code that's not supported in free format) or for declaring procedures with the still fixed format P spec.

It was quite a long wait between free form logic and free form declarations, but they finally arrived in V7.1 (not at the beginning of V7.1 - the support came in at the same time as TR 4, in May 2012). This removed the requirement for /Free and /End-Free - you can still mix fixed and free format code, but no directive is required.

A few restrictions still existed with the 7.1 support - most notably that I and O specs are not replaced (but since they are rarely used in modern coding, it's rarely an issue) and the RPG code was still required to be between columns 8 and 80.

With the new support in V7.2 (and PTF'd back to V7.1) the 8 to 80 column restriction can also go away if you want. If the first statement in a source member is **Free, then the code can be in any column and can extend for as long as the source member or file allows. This support became available in November 2015.

One restriction with the **Free support should be noted that was not true for any of the earlier free-format releases - a source member that has **Free cannot contain any kind of non-free-format RPG code in it. If non-free-format RPG is required for the program, it can be /copied in from an external member at compile time. Effectively data and file declarations in each source member are required to be in a single format, either fixed or to begin with **Free.

We'll Cover Free-Format in Phases

Partner400

1. **A quick overview of free format logic - pre-V7**
   - Syntax rules and op-codes not supported
     - Tips for replacing those unsupported op-codes are at the end of the handout
   - New things you can do only with free-format logic
2. **Free format declarations - V7.1 TR4+**
   - Ctl-Opt - the new H spec
   - Dcl-F - the new F spec
   - Dcl-xx - the new D spec
   - Dcl-Proc - the new P spec
3. ****Free format (no column restrictions) - V7.2 TR 3+ or V7.1 TR 11+**
   - Syntax rules
   - How to overcome the "no fixed format specs" restriction
4. **Converting from fixed form to free form**
   - And tools to help

Notes

We'll be covering free format RPG syntax in historical sequence - that is, free format logic first, followed by free format declarations and lastly the completely free format variant - with no column restrictions.

During the syntax discussion, we'll be assuming we're talking mostly about writing new code in free format. Lastly we'll talk about converting existing code from fixed format to free format, including some tools which you'll want to look into using - because doing it manually is no fun at all!

Free Form Logic - V5R1 +

Completely Free-Form Logic

- Freeform specs originally had to be preceded by a /FREE directive
  - /END-FREE was used to return to fixed format
    - /Free & /End-Free are No longer required as of V7.1 - even if mixing free-form & fixed form
  - As of V7.1 you simply need to leave columns 6 and 7 blank
- You can also mix Free-form and Fixed .... but please don't!

Syntax: Op-Code{(Extender)} { Factor1 } { Factor2 } { Result Field };

- Operands no longer limited to 14 characters
- Each statement must end with a semicolon (;)
- Not all Fixed form Op-codes are supported

Two Op Codes are optional: Eval and CallP


```
/Free
   If CustomerNumber = *Blanks;
      Eval CustomerError = True;
    Else;
      CustomerError = False;
   EndIf;
/End-Free
```

Notes

Although freeform can be mixed with traditional C specs, the result is code that is ugly and hard to maintain. Avoid it like the plague! Note that the sample code on this page has /Free and /End-Free in light gray font. That as required prior to V7.1 but since then, it is not required (and should be removed - just making the code look messy!) Some later examples here will not include those directives.

Wherever possible you should replace operations that do not have a direct freeform equivalent with options that will work in free form instead. The degree to which such replacement is "possible" depends on which release of the operating system you are writing for. We will be discussing the replacement options later in the session.

One of the really nice aspects of using Free-form form calcs is that you can indent your source statements to highlight the logic flow as shown in this short example.

Note the use of the semicolon (;) to specify the end of the statement. As you will soon discover, if you forget it, you will get some really interesting error messages!

As shown in the example, some op-codes can be omitted completely. In the example we have omitted the EVAL op-code In the assignment CustomerError = False. Since the EVAL is optional, either option will work. Later we will look at the exception to this rule.

By the way - the only "Convert to Free-form form" option supplied by IBM is provided within the Rational Developer editor. And that conversion is a very rudimentary one and will only convert logic - no conversion of H, F, D and P specs. Other software vendors, such as Linoma Software and Arcad Software offer much more full-function conversion options for a fee. Craig Rutledge offers an open source (free) tool which converts to free format, including declarations. It involves multiple steps and a fair bit of manual effort and there is no RDi plug-in for it currently.

Free-form Logic Rules

Opcode & operands can appear anywhere in columns 8 - 80

- Positions 6 & 7 must be blank

Comments are specified by the traditional * in column 7

- Or by // in column 8 onwards
  - These // Style comments can be placed at the end of any free form line

There is no way to define fields in Free-form form logic

- But then you wouldn't want to do such a naughty thing anyway!

Resulting indicators may not be specified

- BIFs such as %FOUND, %EOF, %ERROR must be used instead

Embedded SQL was also added to Free-form as of V5R4

```
/Free

 // I can have a comment like this anywhere
EXEC SQL SELECT  NAME, POS
     INTO   :Name, :Job FROM  EMPL
     WHERE  NBR = :EmpNbr;   // Or at the end of a line
```

Notes

If you are not in a position to move to V5's Free-form format yet, there are a number of steps you can take to prepare yourself for the change. These are all good programming style anyway so you can't lose!

Use only those opcodes supported by freeform

Defining all variables on D-Specs

Do NOT use conditioning indicators

Avoid resulting indicators

Note that beginning with V5R4, SQL can be embedded in Free-form RPG logic. Prior to V5R4, many SQL users placed their SQL statements in either subroutines or subprocedures in order to avoid the need for /End-Free to do SQL statements (which themselves were also free format!).

When embedding SQL statements in Free-form RPG logic, the End-Exec directive is not required. And the Exec SQL directive is no longer a "traditional" compiler directive, which would require a slash (/) in position 7 of the spec. Now, the only requirements are simply the words "Exec SQL" at the beginning of the statement and the semicolon (;) at the end of the SQL statement, just as all other Free-form logic statements end.

Free-form Logic Example (V7.1+)

Note that subroutines can be defined and invoked in Free-form

- And as of V7.1 /Free and /End-Free are no longer required

```
READ(E) TransFile;          // Read file to prime do loop

DOW not %EOF(TransFile);    // Continue until everything processed
  IF %Error;
    DSPLY 'The read failed';
    LEAVE;
  ELSE;
    CHAIN(N) CustNo CustMast CustData;
    CustName = %Xlate(Upper : Lower : CustName);
    EXSR CalcDividend;
    READ(E) TransFile;
  ENDIF;
ENDDO;

BEGSR CalcDividend;
  TotalSales = %XFoot(MthSales);
  EVAL(H) Dividend = TotalSales / 100 * DivPerc;
  Record_transaction();
ENDSR;
```

Notice the use of // for end of line comments

Note: EVAL is required here because of the Extender (H)

Notes

Note again the benefit of indentation of the source. With the source coded this way, it is very clear which statements belong to the If block and the Else block. And it is clear that that the IF/ELSE block is all part of the DOW.

Notice that we must use the %EOF and %Error built-in functions because resulting indicators cannot be used in freeform calcs. That's OK, because even in fixed format source, the use of the BIFs makes the code far more readable and understandable.

Notice that in this code sample, the EVAL operation code has been omitted - except where it was necessary to include it because of the half adjust extender.

Now for a tough question: Assuming these calcs are syntactically correct (i.e. the program containing these calcs will compile), what could Record_transaction() be? Is it an array element? A subprocedure call? A program call?

It is either an subprocedure call or a program call (or, to be more specific, it is the name on the prototype given to a subprocedure or program to be called.) The CALLP (call with prototype) operation code, like the EVAL, can be omitted, as it is in this case.

In a free form spec, it is required to use the empty parentheses when no parameters are being passed.

Op-Code Replacements (1 of 2)

Op-Code	Free-form Substitute
ADD	Operator +
ADDDUR	Operator +
ALLOC	BIF %ALLOC
ANDxx	Logical operator AND
BITxx	(Bitwise operations in V5R2)
CABxx	(see GOTO)
CALL	Op-code CALLP
CALLB	Op-code CALLP
CASxx	Op-code IF with EXSR
CAT	Operator +
CHECK	BIF %CHECK
CHECKR	BIF %CHECKR
COMP	Operators =, <, >, etc.
DEFINE	LIKE or DTAARA on D-Spec
DIV	Operator / or BIF %DIV

Op-Code	Free-form Substitute
DOUxx	Opcode DOU
DOWxx	Opcode DOW
END	Opcodes ENDDO, ENDIF, etc.
EXTRCT	BIF %EXTRACT
GOTO	LEAVE, ITER, LEAVESR, etc.
IFxx	IF
KFLD	(See KLIST)
KLIST	(Various options in V5R2)
LOOKUP	%LOOKUPxx or %TLOOKUPxx
MxxZO	(Bitwise operations in V5R2)
MOVE	EVALR or %DATE, %TIME, etc.
MOVEA	%SubArr in V5R3 or …..
MOVEL	EVAL or BIFs for Date, Time etc.
MULT	Operator *
MVR	BIF %REM

Op-Code Replacements (2 of 2)

Op-Code	Free-form Substitute
OCCUR	BIF %OCCUR
ORxx	Operator OR
PARM	(see PLIST)
PLIST	D-Spec PI & PR definitions
REALLOC	BIF %REALLOC
SCAN	BIF %SCAN
SETON	EVAL *Inxx = *ON
SETOFF	EVAL *Inxx = *OFF
SHTDN	BIF %SHTDN
SQRT	BIF %SQRT
SUB	Operator -
SUBDUR	Operator - or BIF %DIFF

Op-Code	Free-form Substitute
SUBST	BIF %SUBST
TAG	(see GOTO)
TESTB	(Bitwise ops in V5R2)
TESTN	(Bitwise ops in V5R2)
TESTZ	(Bitwise ops in V5R2)
TIME	Time & Timestamp BIFs
WHENxx	Opcode WHEN
XFOOT	BIF %XFOOT
XLATE	BIF %XLATE
Z-ADD	Opcode EVAL
Z-SUB	Opcode EVAL

Op-codes not supported in Free-form Logic

These fall into 5 main categories:

1) "Old fashioned" Op-codes whose use is discouraged

- e.g. ANDxx, COMPxx, DEFINE, DOxyy, GOTO, and IFxx

2) Those for which new support has been provided

- e.g. ALLOC, CHECK, EXTRCT, LOOKUP, TIME, XLATE, KFLD, KLIST

3) Op-codes with existing (if not identical) expression support

- e.g. ADD, DIV, MULT and SUB

4) Those supported by a combination of enhanced and new support

- e.g. ADDDUR and SUBDUR

5) The ones the compiler writers don't like: MOVE, MOVEL, MOVEA

- We will talk more about Why they don't like it in a few minutes
- EVAL is the primary alternative but does not meet all needs

Notes

Over the next few charts we will study the substitution options for some of the op codes that are not supported in Free-form format.

Note that the Op-code substitutions are not necessarily one-to-one. The substitute might not work exactly as the original code, especially in the area of error handling. For example the default for numeric overflow on an ADD operation is to truncate the result and ignore the error. The default for an addition in an EVAL type operation is to blow up! You need to bear this in mind if you are converting. Personally we would rather know if numeric overflow occurs, which is why we have always preferred using EVAL to the older style operations.

What's So Bad About MOVE ?

It tells you (almost) nothing about what is happening

For example - what does this code do?

```
C                   MOVE      ARCode        PrARCode
```

Does it ... ?

- Simply copy the data - byte for byte?
- Change the data type?
  - And if so what other effect did that have?
- Truncate the data?
- Effectively concatenate two fields?
- Discard high-order digits without warning ?

And how about this ?

```
C     *MDY          MOVE      InvDate       WorkDate
```

- Which is the "real" date field (i.e. the D data type) ?
- Which one is in MDY format ?

Notes

For many people the surprise omission from the list of op-codes supported in Free-form was MOVE. There were also a few people bent out of shape over the omission of GOTO - but that has been accepted as poor programming practice for so many years that they received little support from their fellow programmers.

MOVE on the other hand caused a lot more controversy. As a result, Internet flame wars raged on and off for several years on this topic with the pro and con MOVE camps fairly equally divided. However, IBM stuck by their guns and now some years later still show no sign of changing their minds. We tend to lean towards the "MOVE is evil" camp for reasons that we hope will be apparent from the chart above.

Replacing Call - PRrototype

Partner400

The modern way to call anything

- e.g., program or procedure or subprocedure

Parameter mismatches can become a thing of the past!

- The compiler can validate your program calls
- Prototypes allow the compiler to validate:
  - The number of parameters and
  - Their data type and size

Prototypes can accommodate differences in data definition

- For example, allows an integer to be passed when the callee expects a value with decimal places, allows an expression to be a parameter

The op-code CALLP is used with Prototypes

- It means CALL using Prototype (not Call Procedure)

Free format version of prototype declarations will be covered later

```
D TaxCalc         PR                  ExtPgm('PX027C')
D  GrossAmount                 8S 2  CONST
D  AllowanceAmount             8S 2  CONST
D  TaxAmount                   8S 2

   TaxCalc( Gross : Base + PersAllow : Tax);
```

Notes

Prototypes can be used when calling anything - a program, whether RPGLE or RPG or CLP or CLLE, etc. or a subprocedure or a C function, a system API - anything.

Prototypes were added to the RPG IV language in the V3R2 and V3R6 releases. Although their initial purpose was to support Subprocedures, the RPG developers realized that they could be used to provide additional support for program calls among other things. This helps to address one of the really annoying problems with the old CALL/PARM syntax. Namely that errors with parameter lists are not discovered until run-time. It would be much better if we could have the compiler validate the parameter lists for us, and that's what prototyping provides.

In addition to validating parameters, prototyped calls allow for automatic adjustment in the case of certain parameter mismatches. For example they can allow you to pass a 7 digit packed field when a 9 digit zoned is expected. The techniques involved are beyond the scope of this session but some references to more information about prototypes are included below.

A great many people mistakenly believe that prototypes and CALLP relate only to procedures. This is not true. CALLP can invoke either a program or a procedure. As we shall see shortly, which type of call is made depends on keywords on the PR line of the prototype.

If you need to know more about using prototypes, take a look at these articles:

RPG IV Prototypes: What are they and why should you care? - Susan Gantner

http://www.ibmsystemsmag.com/ibmi/developer/rpg/iSeries-EXTRA----RPG-IV-Prototypes--What-are-they/

Parameters and Prototypes - Scott Klement

https://www.scottklement.com/presentations/Parameters%20And%20Prototypes.pdf

The Geezer's Guide To Free-Form RPG, Part 4: Prototypes and Procedure Interfaces

http://www.itjungle.com/fhg/fhg070914-printer01.html

Now Appearing in Free-form Logic Only

Notes

I/O Enhancements - New Key Options

%KDS is a free-form replacement for KLIST

- Used in Factor 1 position instead of a KLIST name

Syntax:

%KDS(keyDSName { : numberOfKeysToUse })

- Optional 2nd parameter specifies # of fields to be used in partial key

All keyed operations can use this new BIF

- Generate related DS by using LIKEREC(recname : *KEY)

But don't get too excited ... there's an even better option coming up ...

```
  // The ProductKeys DS will contain the three fields that make up the
  //   key for the Product File (Division, PartNumber and Version)
D ProductKeys     DS                  LikeRec(ProductRec : *Key)

// Read specific record using full key
   Chain %KDS(ProductKeys) ProductRec;
// Position file based on first 2 key fields
   SetLL %KDS(ProductKeys : 2 ) ProductRec;
```

Notes

Until this feature was introduced, there were two ways to specify the key for a keyed operation. Specify the name of a single field or the name of a KLIST. KLISTs always annoyed me because you had to wander off elsewhere in the program to actually find the list. Only then did you know what keys were being used. This new support offers both an improved alternative to the KLIST approach and a new method of directly specifying the keys on the operation itself.

The new "KLIST" (actually a BIF called %KDS - Key Data Structure) references key definitions in the D specs where they belong. Remember the LIKEREC *KEY option we covered earlier? This is where it comes into play. You can use it to automatically generate a DS containing the file's key fields. This structure can then be referenced in the I/O operation by specifying the DS name to the new %KDS function.

So how do you specify that a partial key is to be used? Just use the second parameter of %KDS to tell the compiler how many of the key fields are to be used.

We will look at the second method on the next chart.

I/O Enhancements - New Key Options

Partner400

Keys can now also be specified as a list of values

- List is specified in Factor 1 position, enclosed in parentheses

Any type of field, literal or expression can be used

- As long as the base type matches AND the key is in parentheses
  - i.e. The result is numeric for a numeric key, Alpha for an alpha key, etc.
- The compiler will perform any required conversions
  - Using the same rules as for EVAL operations - much like CONST does for parameters
- Note that the absence of parentheses changes the behavior
  - See example below - Rule of thumb - use parentheses even for

```
// Read using specified keys
 Chain (Division : PartNumber : Version) ProductRec;

// Position file based on first 2 key fields
 SetLL (Division : PartNumber) ProductRec;

// Position file to specified Part number in Division 99
 SetLL ( '99' : PartNumber) ProductRec;

 Chain Field Record;   // requires an exact match of attributes for Field
 Chain (Field) Record; // only requires a matching base type
```

Notes

The second method is an extension of the current ability to specify a single field as the key (the old Factor 1).

Instead of a single field, you can now supply a list of fields. The list should be specified within parentheses with colons (:) used to separate the individual key elements.

Note that the key elements do not have to be fields, they can be any character expression. The compiler will perform conversion if required.

Note that even if you have a single field key, it is often better to enclose it in parentheses because with the parentheses the field specified is not required to match exactly in type and size, but without parentheses they must match. This is much like the size & type accommodation made by the compiler with the use of the CONST keyword in prototypes for parameters.

I/O Enhancements - %FIELDS

Partner400

Allows you to specify a list of fields to be affected by an UPDATE

- Only those fields specified will be updated

Syntax:

%FIELDS(name1 { : namen ... })

Used in the Result position of the op-code

```
// Update all the fields in the Product record
 Update Product;

// Update only UnitCost and UnitPrice fields
 Update Product %Fields( UnitCost : UnitPrice );
```

Notes

In many ways this is one of the best of the new features, and we saved it for the end (almost)

This is the capability to limit which fields are modified by an UPDATE operation. We love this one! It provides a great way to protect your code from the worst efforts of (shall we say) less-gifted programmers. The list of fields is specified using the new BIF %Fields. Only those fields specified will be updated.

Why is this useful? Suppose that, during the operation of the program, only certain fields in the file should be subject to change. By specifying those fields to the UPDATE op code, you are assured that only those fields will be changed. If during subsequent maintenance tasks a mistake is made and the value of a field that should not be modified is accidentally changed in the code, it will have no effect on the database. Only if the %Fields list is also modified can this error result in database corruption.

Short-form Expression Support

Partner400

I suspect you'll either love this or hate it

- It is kinda nice when using long or subscripted or qualified names
  - Particularly if you are a slow typist !!

It allows shortening expressions such as X = X + 1 to X += 1

- Instead of repeating the field name (X) you can use short-forms
  - += for addition
  - -= for subtraction
  - and you can guess what the others are ...

This is not limited to Free-form

```
// This calculation
     MonthTotal = MonthTotal + TotalSale;
// can be shortened to this
     MonthTotal += TotalSale;
// and this
     x = x + 1;
// can similarly be shortened to this
     x += 1;

// Be careful! The following is not the same
     x =+ 1;    // This is the same as x = 1;
```

Not strictly a free-form feature but is more often used in free-form logic

Notes

Numeric operations now support short-form notation for certain functions. Prior to this release, an addition of the type X = X + 1 required that you repeat the name of the target field. Some people considered this a step backwards since the old ADD op-code offered a short-form notation that only required the target field to be specified once, in the result field. e.g. ADD 1 X.

With this new feature, the expression can be written as X += 1. Similar shorthand can be used for subtraction, multiplication, division, and exponentiation.

Do not make the mistake of coding it like this: X =+ 1 This simply puts a positive value of 1 into X.

V7+ Free-Form Declarations

Now (almost) all RPG code can be free-format
- Control (aka H) specification
- File Declarations
- Data Declarations

- Procedure & Prototype declarations



Notes

Beginning with V7.1, nearly all the RPG language can be coded in free format. The only exceptions are the I and O specs but since those are so rarely actually coded in RPG IV, that is not much of a limitation.

This section will describe the details of replacing control specs, file specs and data declarations specs with the new free format statements.

We have not yet covered subprocedures or prototypes so free format statements to replace those will be covered later.

IMPORTANT NOTE: The SEU editor (aka the green screen editor accessed via PDM) does NOT support any of these new 7.1 free format versions of RPG. So you can code them using SEU but all the code will be indicated as in error because support for that syntax doesn't exist - and will not. The editor in the RDi (Rational Developer for i) product is the only editor from IBM that supports the V7+ RPG syntax.

V7 - TR7 Free Form Enhancements

Partner400

You can now specify an entire program in free form

- Only exception is that I and O specs are not available in free-form

No longer any need for /Free or /End-Free

- Just leave columns 6 and 7 blank

New free form options for:

- H-specs (CTL-OPT)
- F-specs (DCL-F)
- D-specs (DCL-xx)
  - Where xx = C, DS, PARM**, PI, PR, S, or SUBF (These are rarely needed)
- P-specs (DCL-PROC)

Not Technically a part of TR7 but was announced and released at the same time

```
Ctl-opt option(*srcstmt) dftactgrp(*No);

Dcl-ds employeeDS;                      // Begin Data Structure
  firstName char(16)    Inz('James');   // Subfields
  lastname  char(30)    Inz('Joyce');
  salary    packed(7:2) Inz(12500);
End-ds;                                 // End Data Structure

Dcl-f qprint printer(80);  // Define PRTF and associated DS together
Dcl-ds prtDs len(80) end-ds;
```

Notes

For many RPGers - especially those who use subprocedures - one of the best parts of this support is the ability to do away with /Free and /End-Free!

But the ability to actually code a file declaration without your secret decoder ring to tell you what character to put in what column is also very nice!

Another feature is illustrated in this example - the ability to intermix file declarations with "ordinary" (that is, D-spec-type) declarations. So files and standalone fields, data structures, etc may be in any order - but all must come before the logic.

But First ... Let's Talk about Editors

Free-form Declarations are not supported in SEU/PDM editor

This is what V7+ free-form coding looks like in SEU

- IBM will NOT be enhancing SEU to handle these new definitions

It's time to move up to a modern editor - e.g., RDi



Notes

IBM has confirmed that SEU will not be updated to support this new functionality. In fact it was frozen as at V6 and supports none of the V7 functionality.

As a result a great many people have started doing what they should have done many years ago and moved their development to RDi. It fully supports the new free-form including excellent code-assist to help you remember the new formats.

Some other editors not from IBM also support free format syntax - such as Liam Allan's ILEditor - not nearly as full-function as RDi but still light years ahead of SEU.

RDi Fully Supports Free Format Coding



Notes

Note that not only does RDi not flag all my free-form declarations as errors, I also allows me to see much more lines of code on the screen at a time (typically 2 to 3 times more) and - just for fun - I've also included the Outline view which shows the definitions, including all the externally described fields from the files.

Can't Recall that Keyword?

RDi's Content Assist (Ctrl + Space) to the rescue



Notes

RDi's Content Assist support is particularly helpful when you first start coding with free-form declarations.

For example, it can bring up a list of keywords that are valid in the context where your cursor is - so if you can't remember the keyword to define a Date or an Integer or a Packed Decimal field - just ask for content assistance. It's not just for keywords on data types, it works for all types of keywords. It even works in logic statements

Format of the new Declarations

Partner400

The new declaration op-codes follow this basic format:

- First, the DCL-xx itself
- Next the name of the item
  - File, field, procedure, etc. (no name for Ctl-opt)
- Followed by keywords related to other fixed form elements
  - e.g. File usage, field type and length
- Then keywords from the old spec

A more complete code sample is on the next chart

```
ctl-opt option(*srcstmt) dftactgrp(*No);

dcl-ds employeeDS;
  firstName char(16) Inz('James');
  lastname  char(30) Inz('Joyce');
  salary    packed(7:2) Inz(12500);
end-ds;

dcl-f qprint printer(80);

dcl-ds MyExtDs ExtName('MYFILE') end-ds;
```

From original D-spec keyword area

Fixed column entries

end-ds can be on the same line as dcl-ds in this case.

Notes

When it comes to replacing D specs, DCL-xx includes:

DCL-S - Stand-alone field
DCL-DS - Data Structure (usually requires END-DS)
DCL-C - Named constant
DCL-PR - Prototype (usually requires END-PR)
DCL-PI - Procedure Interface (usually requires END-PROC)

Later, we'll see that we can also replace P specs with:

DCL-PROC - Define a procedure (requires END-PROC)

In all cases where an END-xx is allowed, the name of the DS/Prototype/Procedure may be included on the END-xx

Simple Free Form Program

File definitions can now be intermixed with data definitions

- Named Constants, Data Structures, Stand-alone fields, etc.

Most new options have sensible defaults - Less coding required!

- Disk files default to input, Printers to output, Decimals to zero, etc.

End of line comments are now useful in definitions!

```
ctl-opt option(*srcstmt) dftactgrp(*No);

dcl-ds employeeDS;  // Nice to be able to have comments here!
  firstName char(16) Inz('James');
  lastname  char(30) Inz('Joyce');
  salary    packed(7:2) Inz(12500);
end-ds;

// Define printer file and associated DS
dcl-f qprint printer(80);     // This PRTF is program described
dcl-ds prtDs len(80) end-ds;

dsply ('Hello to our new employee');
dsply ( %TrimR(firstName) + ' ' + lastName );

prtDs = 'The name of our new employee is ' +
         %TrimR(firstName) + ' ' + %TrimR(lastName) +
         ' his salary is $' + %Char(salary);
write qprint prtds;
```

Note that Printer file is defined along with the DS that is used for its output

Notes

CTL-OPT - The New H-Spec

Defaults to DftActGrp(*No) if any ILE-specific options are used

- ACTGRP, BNDDIR, or STGMDL
  - This is the first of many "sensible" defaults that are added by this support

Other differences:

- Can start in any column from 8 onwards
  - Exception: In **Free members, can begin before column 8
- Must be terminated with a semi-colon

No other differences between this and the current H-spec

- Can occupy multiple lines
- Multiple CTL-OPT can appear in program
- Presence of CTL-OPT stops compiler from looking for H spec data area

```
H BndDir('UTILBNDDIR') Option(*SRCSTMT: *NODEBUGIO) DftActGrp(*No)

   // May now be coded as:

  Ctl-Opt BndDir('UTILBNDDIR');
  Ctl-Opt Option(*SRCSTMT: *NODEBUGIO);

   // Can also be coded as:

  Ctl-Opt BndDir('UTILBNDDIR') Option(*SRCSTMT: *NODEBUGIO);
```

DftActGrp(*No) not needed - Assumed due to BndDir

Notes

Not that much has changed about the H spec since it was almost free format before.

Ctl-Opt replaces the H and the semi-colon is used at the end.

Note that just as the H spec would prevent the compiler going to look for an H spec data area, so will the Ctl-Opt statement. If you just want to prevent the data area search but have no other options to specify, you could simply have an empty statement, such as:

Ctl-Opt;

With all the new free-format definitions and even logic (provided you're not in a **Free member with no column restrictions), you are allowed to intermix free-format with fixed format - although it usually produces an ugly mess, so it's best to avoid it.

Syntactically, the compiler would accept the following in a single member (but please don't do it!):

```
H option(*srcstmt : *nodebugio)
  ctl-opt datfmt(*YMD) nomain;
H timfmt(*usa)
```

Declaring Files In Free-Form

Partner400

File Name listed first, followed by Device type keyword (if any)

- Device type defaults to DISK if omitted - i.e., a Database table

Externally described is the default

- File Keyword *EXT can optionally be specified as a parameter
- Program described files must specify the record length
  - e.g. PRINTER(132) for a program described printer file

No more copying F specs!

Sensible defaults for USAGE based on device type - more details coming

Add KEYED keyword for keyed database (disk) files

- Other F spec keywords may also be used

File name no longer limited to 10 characters

- EXTDESC used to specify actual name when needed (also need EXTFILE)

```
Dcl-F CustMaster Disk  Usage(*Update) Keyed;
Dcl-F Report     Printer(*EXT) Usage(*Output) OFLIND(*IN96);
Dcl-F qPrint     Printer(132) OflInd(PageFull); // Program described
Dcl-F MyDisplay WorkStn Usage(*Input:*Output);
Dcl-F InvMast;      // Defaults to Disk device, Input usage
Dcl-F InvoiceMaster ExtDesc('INVMAST') ExtFile(*EstDesc);  // Long name
```

Notes

In the example here, we have defined several files. We've shown different ways of defining the same type of file, taking advantage of the intelligent defaults.

The OFLIND keyword on the Report file is “Overflow Indicator” which can be used by the program to determine when to go to a new page on the report.

The File name is no longer limited to 10 characters as it was on the F spec. However, since file object names (externally on the system) are limited to 10 characters, that means if a longer name is used, then keyword ExtDesc must be coded to give the real external name for the file. A longer, more meaningful name for the file may be useful for making the logic more readable.

Note that when using file names longer than 10 characters, both the EXTDESC and EXTFILE keywords are required. There is a shortcut in that you can specify EXTFILE(*EXTDESC) to avoid the need to key the file name twice - as shown in the example on this chart.

There are other keywords that can be used when declaring files. These are the most commonly used features of file declarations.

DCL-F : Very Helpful Defaults

Partner400

Device type implies default usage

- **Disk** device defaults to **Usage(*Input)**
- **Printer** device defaults to **Usage(*Output)**
- **WorkStn** device defaults to **Usage(*Input:*Output)**

Some Usage values imply others

- Usage values are: *Input, *Output, *Update, *Delete
  - **Usage(*Update)** implies ***Input**
    - Warning! Unlike with with F spec, *Update does NOT imply *Delete
  - **Usage (*Delete)** implies both ***Input** and ***Update**

```
FCUSTMR0   UF A E           K DISK    USROPN
FREPORT    O    E             PRINTER OFLIND(*IN96)
FSCREEN    CF   E             WORKSTN

   Dcl-F Custmr0 DISK Usage(*Input:*Update:*Delete:*Output) Keyed UsrOpn;
   Dcl-F Custmr0 Usage(*Delete:*Output) Keyed UsrOpn;

   Dcl-F Report  Printer(*EXT) Usage(*Output) OFLIND(*IN96);
   Dcl-F Report  Printer OflInd(*IN96);

   Dcl-F Screen Workstn Usage(*Input:*Output);
   Dcl-F Screen Workstn;
```

Notes

In the free-format version of the example, note that there are 2 versions of each file declaration. In each case, both versions have identical effect. Due to defaults and assumptions made by the compiler, the 2nd version of each is a "minimal" one, taking advantage of defaults and compiler implications.

DISK device type is the default if none is coded. Device type must be the first thing coded after the name of the file, with 1 exception. The exception is when the file is defined using the LIKEFILE keyword. In that case, the LIKEFILE must immediately follow the name.

The most commonly used devices are: DISK, PRINTER, and WORKSTN

There are less common device types: SPECIAL (where a user-created program handles the I/O operations for the file) and SEQ (where the file is sequentially organized - the actual device is specified in a CL command or in the file description).

It is important to note that in fixed form, Update implies that the files is also delete-capable. However, in free-form, *DELETE must be explicitly specified for the files to be delete-capable.

DCL-xx - The New D-Spec

xx = DS for Data structures

- In most cases there must also be a matching END-DS

xx = SUBF for DS subfields - Very Rarely Required

- Code only if field name is a valid free-form op-code
  - Yes, some strange people do use names like READ or SELECT as field names

xx = S for Stand-Alone fields

xx = C for Named Constants

```
D Address         DS                  Dim(20) Qualified
D  Street1                          30A
D  City                             30
D  State                             2
D  Zip                               5S 0
D  ZipPlus                           4S 0

  dcl-ds Address Dim(20) Qualified;
    Street1    char(30);
    City       char(30);
    State      char(2);
    Zip        zoned(5);  // Zero decimals assumed
    ZipPlus    zoned(4:0);
  end-ds Address;
```

DS Name optional at end
If present must match

Notes

END-DS may be omitted if DS is externally described or has no named subfields. It can even be on the same line as the DCL-DS in such cases, such as the following DS which is externally described based on a table (aka file) named PRODUCT.

```
dcl-ds product          Ext end-ds;
```

Name of DS can be specified on end-ds and must match if present.

DCL-xx - The New D-Spec - Data Types

Data types are spelled out instead of using a one character code

- Chart below shows the most commonly used types
- Length or format of item follows in parentheses when required

Simpler definition for some data types

- VarChar avoids the need for the Varying keyword
- Date, Time, Timestamp - no need for separate DatFmt keyword
- 0 (zero) decimals assumed for all numeric data types if not specified.

Data Type	Free Keyword	Comment
A	Char(len)	
A + Varying	Varchar(len)	
I	Int(len)	Decimals not specified
U	Uns(len)	Decimals not specified
P	Packed(len:dec)	0 assumed if decimals not given
S	Zoned (len:dec)	0 assumed if decimals not given
N	Ind	
D + DatFmt	Date(format)	
B	BinDec (len:dec)	!! DO NOT USE !!

See Notes for examples

Notes

For the answer to the question "Why should we not use BINDEC?" see this article http://ibmsystemsmag.com/ibmi/developer/rpg/binary-integer/.

The following is a partial list of RPG data types represented as D-specs:

```
d packedNum       s              7p 2
d zonedNum        s              7s 2
d integer         s             10i 0
d unsigned        s             10u 0
d float           s              8f
d character       s             20a
d varyingChar     s             20a   Varying
d dateMDY         s               d   DatFmt(*MDY)
d timeUSA         s               t   TimFmt(*USA)
d indicator       s               n
d nastybinary     s              9b 0
```

And here are their free form equivalents.

```
Dcl-S packedNum    Packed(7:2);
Dcl-S zonedNum     Zoned(7:2);
Dcl-S integer      Int(10);
Dcl-S unsigned     Uns(10);
Dcl-S float        Float(8);
Dcl-S character    Char(20);
Dcl-S varyingChar  Varchar(20);
Dcl-S dateMDY      Date(*MDY);
Dcl-S timeUSA      Time(*USA);
Dcl-S indicator    Ind;
Dcl-S nastybinary  Bindec(9);
```

D-Spec DS Example

Partner400

This shows how a fixed form DS would be converted to free-form

- I have used my own personal preferences for alignment
  - At least they were my preferences at the time I coded this example!

```
D MyDS            DS
D  Address                        32A
D    Street                       15A   Overlay(Address)
D    City                         10A   Overlay(Address: *Next)
D    State                         2A   Overlay(Address: *Next)
D    Zip                           5A   Overlay(Address: *Next)
D  LstOrdDate                      D   DatFmt(*USA)
D  ASubfield                      25A   Varying

Dcl-DS MyDs;
  Address     Char(32);
    Street      Char(15) Overlay(Address);
    City        Char(10) Overlay(Address:*Next);
    State       Char(2)  Overlay(Address:*Next);
    Zip         Char(5)  Overlay(Address:*Next);
  LstOrdDate  Date(*USA);
  ASubfield   Varchar(25);
End-DS MyDs;
```

Notes

I like the idea of aligning the field type definitions even though it is not required. I don't however insist on placing them in a specific column. Rather I start them 2 characters after the end of the longest field name in the block. For me it works but you'll devise your own style - just be consistent and remember that readability is critical.

D-Spec - Unnamed Fields and POS

In fixed form no field name was required

- In free-form we must specifically inform the compiler
  - Done by coding *N instead of a field name

In free-form we are not permitted to OVERLAY the DS name

- We must use the new keyword POS(1) instead
  - POS(nn) can also be used for positioning any field within a DS
  - For example when specifying specific fields in the PSDS

```
Dcl-DS DayData;

  *n         Char(9) inz('Monday');    // *n = no name
  *n         Char(9) inz('Tuesday');
  *n         Char(9) inz('Wednesday');
  *n         Char(9) inz('Thursday');
  *n         Char(9) inz('Friday');
  *n         Char(9) inz('Saturday');
  *n         Char(9) inz('Sunday');

  DayArray   Char(9) Dim(7) Pos(1);

End-ds;
```

Fixed format version of this DS is shown in the notes

Notes

The RPG compiler team decided that too many people got confused when the OVERLAY keyword was used against the DS name. To help avoid this confusion, OVERLAY is now limited to subfields within the DS and you must use the POS keyword to reference the DS starting position.

This is effectively the same as using a start position on the old D-specs.

This is what the DS in my example would have looked like in fixed-form. Another good use for POS is with Indicator Data Structures (INDDS).

```
D DayData          DS
D                                9    Inz('Monday')
D                                9    Inz('Tuesday')
D                                9    Inz('Wednesday')
D                                9    Inz('Thursday')
D                                9    Inz('Friday')
D                                9    Inz('Saturday')
D                                9    Inz('Sunday')

D DayArray                       9    Overlay(DayData) Dim(7)
```

2 Ways to Code a Nested DS

Partner400

```
Dcl-Ds Divisions Qualified Dim(5);
   DivCode     Zoned(2);
   Departments LikeDS(DeptData) Dim(10);
End-Ds;

Dcl-Ds DeptData Qualified TEMPLATE;
   DeptCode    Zoned(3);
   Products    LikeDS(ProductData) Dim(99);
End-Ds;

Dcl-Ds ProductData Qualified TEMPLATE;
   ProdCode    Zoned(5);
   MonthsSales Packed(9:2) Dim(12);
End-Ds;
```

Option 1:
Use LikeDS

```
Dcl-Ds Divisions Qualified Dim(5);
   DivCode     Zoned(2);
   Dcl-DS Departments Dim(10);
      DeptCode    Zoned(3);
      Dcl-DS Products    Dim(99);
         ProdCode    Zoned(3);
         MonthsSales Packed(9:2) Dim(12);
      End-DS Products;
   End-DS Departments;
End-Ds Divisions;
```

Option 2:
Inline Coding
Only in Free-form

Inline nested DS coding is proof that the RFE process works!

Notes

Nested Data Structures are not new for free form. However, there are 2 ways to code them in free-form.

The first way is just free-format version of what we did with the D specs using the LikeDS keyword. That's shown at the top of this chart.

The second way is a more direct form of nesting - where you actually code the nested DS inline and omit the LikeDS.

This second form was implemented as an RFE - Request for Enhancement - on IBM's developerWorks site. So when you come up with good ideas for enhancements to RPG or most any other IBM software product, submit it to the RFE site and get others to vote for it. It just may be implemented like this one was.

More Fun with Data Definitions

Constants can be used in many more places

- Including field length, decimal places, array dimensions - just about anywhere you would use a literal

Long names don't require ellipsis … or a continuation line

```
D DateMDY         S               D   DatFmt(*MDY-)
D array           S              10a  dim(10)
D CustomerInfo    DS
D   CustomerName                 50A  Varying
D   CustomerBalance...
D                                 7P 2
```

```
dcl-s DateMDY           Date(*MDY-);
dcl-s array             Char(10) Dim(arrayMax);
dcl-c  Digits 7;
dcl-c  Decimals 2;
dcl-ds CustomerInfo;
  CustomerName     VarChar(50);
  CustomerBalance  Packed( Digits: Decimals );
end-ds;
```

Notes

Note that in the fixed format example here, it made no sense to define Digits and Decimals as constants since they could not be used to define the length and precision in fixed format D specs. However, in the new free format D specs, they can be used - as illustrated in the 2nd example.

Note that not only are the ellipsis not required for longer variable names - they are not allowed in most cases. Unless the actual variable name is continued on the next line, ellipsis should not be used, even if the data type and other related keywords are on the next line. If ellipsis are found, the compiler assumes the next line will contain part of the variable name.

Subprocedures, Prototypes and More

Partner400

DCL-PROC declares a subprocedure

- Completed by an END-PROC {name}
- Name is optional but if included must match DCL-PROC name

DCL-PI and DCL-PR define procedure interfaces and prototypes

- Placeholder *N can be used if you don't want to type the name again in PI
  - Or for unnamed parameters in a prototype
- DCL-PARM within PIs & PRs is like DCL-SUBF within DS
  - Only needed if name of parameter is the same as an RPG op-code
  - Such as a parameter named SELECT or CHAIN, as in

    Dcl-Parm SELECT Char(2);

```
dcl-proc DayOfWeek Export;

dcl-pi *N Int(3);              // To omit name, use *N placeholder
  InputDate Date(*USA) Value;
end-pi;

  dcl-s DayNumber int(3);

  // Do logic leaving value in DayNumber
  Return DayNumber;

end-proc DayOfWeek;
```

Notes

Note that the syntax rules for defining parameters in prototypes (PR) or procedure interfaces (PI) are very similar to those for defining subfields in a data structure (DS). The DCL-PARM is not required unless the parameter name is the same as an RPG operation code.

Also, similar to the DCL-DS and END-DS, the name on the END-PROC, END-PR, END-PI is optional but if specified it must match the one at the beginning (of course!).

Prototypes (PR) and Procedure Interfaces (PI)

Program name can be omitted from EXTPGM

- If the program name is the same as the prototype name
- As before, the name, if specified, name must be upper case and in quotes

New option for EXTPROC - *DCLCASE

- Indicates the external name is exactly the same, including case
  - Avoids need to retype name when using mixed case procedure names
    - such as with mixed-case APIs or C functions

All the PRs and PIs here are incomplete - no parms and no End-xx

```
dcl-pr QCmdExc ExtPgm ; // Can be used to call 'QCMDEXC' program

dcl-pr QCmdExc ExtPgm('QCMDEXC') ; // Can also call 'QCMDEXC' program

dcl-pr MyProgram ExtPgm; // Can be used to Call 'MYPROGRAM'

dcl-pr DifferentName Extpgm('MYPROGRAM'); // Call 'MYPROGRAM' using the
                                          //   name DifferentName

dcl-pr cos float(8)  ExtProc(*DclCase);  // Calls C function 'cos'

dcl-pr cosine float(8)  ExtProc('cos');  // Also calls C function 'cos'
```

Notes

Note that the *DCLCASE option is only available for EXTPROC, not for EXTPGM - for hopefully obvious reasons!

As the note indicates on the examples, these are all examples of just the first part of the declaration of PRs or PIs. Obviously most PRs and PIs have parameters and all of them require an END-PR or END-PI as well. We just didn't have enough room on the chart to show all the variations we wanted to illustrate if we had to include parms and END-xx as well.

If there are no parms - just a return value - then it is possible to include the END-PR or END-PI at the end of the DCL-PR or DCL-PI, such as the one below. But using a separate End-Pr; statement would also be valid.

```
Dcl-Pr  getCustName  Char(25)  End-Pr;
```

"Gotchas" to Watch Out For

OVERLAY keyword cannot be used against DS name

- Use POS(1) instead

Names in EXTNAME, EXTFLD, and DTAARA

- Must be in quotes and are case sensitive
- Without quotes, they are treated as a variable or constant name

Ellipsis (...) for continuation only allowed when continuing a name

- But almost never needed anymore anyway

On F-Spec "U" enables both update and delete

- In free form *DELETE must be requested explicitly

End-DS, End-PR, End-PI are (almost) always required

- But may appear on same line as DCL-xx in some cases

RDi's "Convert all to Free-Form" means only convert "all logic"

- And will still generate /Free and /End-Free
- And still gives up on anything even slightly difficult to figure out

I and O specs remain in fixed form

- Probably forever

Notes

Here are few things that have caught us and a few others using the new free form declarations by surprise.

If you know of or find additional gotchas, please let us know!

High Points for Free-Form Declarations

Partner400

/Free and /End-Free no longer required

- Even if not using free-form declarations, as of V7.1 not /Free or /End-Free needed

Intelligent defaults

- In file declarations
  - For device type (Defaults to DISK)
  - Usage (Default based on device type)
- In data declarations
  - Decimal places (Defaults to zero)

Check out content assist (Ctrl+Space) in RDi for help with remembering data type names, etc.

File and data declarations can be mixed

- Even in fixed form

Constants can be used as keyword values

- Including for lengths and decimal places

/Copy and other compiler directives no longer need to start in col 7

// style end of line comments can be used in data/file declarations

DFTACTGRP(*NO) is optional

- If any other ILE keyword such as ACTGRP or BNDDIR is used

Notes

They had me at "no more /Free and /End-Free".

But then it's really nice to be able to actually read (and understand) a file declaration. And I love not having to copy an F spec from another program because I could never remember what things went in which columns.

And the intelligent defaults - such as no longer having to specify that I'm not planning on reading anything from my printer file!

If You Want More Columns…

Partner400

If the first line of your source says **Free

- **Free must begin in line 1, column 1 (ironic, isn't it?)
- Then you can continue coding in column 1
- And continue coding all the way to the end of the source line
- Requires V7.3 or V7.2 TR3 or V7.1 TR11
- **Free members cannot contain ANY non-free-format statements
- If fixed-format code is needed, use /Copy to bring it in at compile time
  - Each /Copy member is assumed to be "column-limited" unless line 1 says **Free
- /Free and /End-Free are not only not required - they are not allowed

SQL pre-compiler supports **Free

RDi 9.5 and later supports **Free

- But not SEU, of course

Source members are limited in source lines to 32,766

- That ought to be long enough!
- IFS files are unlimited

Notes

Tools for Converting Old to New

Built-in RDi Option (Source menu) - "Convert All to Free-Form"

- **It doesn't "Convert All"** - not even close
  - Only logic - no declarations
  - Even then, only very rudimentary/obvious conversions are done

Linoma Software's RPG Toolbox

- The latest evolution of the first commercial RPG III conversion utility
  - Includes conversion to Free-form logic, declarations, etc.
  - Command interface + RDi plug-in - allows for partial "selection conversion" in RDi

Arcad Software's ARCAD Transformer RPG

- Converts source members to Free-form - including declarations
  - Command interface + RDi plug-in
  - Even converts some GoTo logic!

Craig Rutledge's open source tools

- JCRCmds free toolset contains some utilities related to Free-form
- Involves multiple steps + some manual effort - but it is free
- New open source RDi plug-in from Thomas Raddatz on Sourceforge

Notes

Don't waste any of your time with IBM's RDi free-form convertor. It was never intended to be a serious "contender" in this arena. It works on a line-by-line basis without any reference to the data types/lengths of the fields being used. As a result it reverts to fixed form for all MOVEs - even if the two fields involved are identical in size and type. This makes for really ugly code. If all of your code is in EVAL style - no MOVEs etc. - then it does an OK job. However, if you are doing a major modernization effort on RPG III style code don't bother with it - just go for one of the alternatives.

Speaking of the alternatives:

Craig's tools are the cheapest - the only cost is your time to build and modify them. I haven't checked out the RDi open source plug-in that Thomas Raddatz has done yet. That should be a great enhancement to Craig's toolset.

Many shops already have Linoma Software's RPG toolbox, but it is fairly inexpensive even if you don't. Adding their extra-cost RDi component is an easy addition to justify.

Last but not least Arcad. Their tooling perhaps the most comprehensive. In fact if you give it a chance it can even rid your code of pesky GOTOs!

Linoma's tool does a great job but ...

- It tends to play it safe and includes a lot of extra keywords
- I'm also not a fan of the way it aligns things - it doesn't!
  - But they are looking at changing this and providing more formatting options




```
000002      Dcl-F Customer Disk(*Ext) Usage(*Input);
000003
000004      Dcl-Pr TaxCalc    ExtPgm('PX027C');
000005        GrossAmount Zoned(8:2) CONST;
000006        AllowanceAmount
000007        Zoned(8:2) CONST;
000008        TaxAmount Zoned(8:2);
000009      End-Pr;
000010
000011      Dcl-Ds myDataArea  DtaAra('JONSDATA');
000012        lastRunDate Date(*USA);
000013        lastSeqNum Int(5);
000014        lastInvNum Packed(7);
000015      End-Ds;
000016
000017      Dcl-S Gross Packed(7:2);
000018      Dcl-S Base Packed(6:2);
000019      Dcl-S PersAllow Packed(5:4);
000020      Dcl-S Tax Zoned(8:2);
000021
000023      *InLR = *On;
```

```
000004     FCustomer  IF   E              Disk
000006
000007     D TaxCalc         PR                   ExtPgm('PX027C')
000008     D  GrossAmount                8s 2  CONST
000009     D  AllowanceAmount...
000010     D                             8s 2  CONST
000011     D  TaxAmount                  8s 2
000012
000013     D myDataArea      ds                   DtaAra(JonsData)
000014     D   lastRunDate                 d      DatFmt(*USA)
000015     D   lastSeqNum                5i 0
000016     D   lastInvNum                7p 0
000017
000018     D Gross           S           7p 2
000019     D Base            S           6p 2
000020     D PersAllow       S           5p 4
000021     D Tax             S           8s 2
000022
000023       *InLR = *On;
```

Notes

My biggest grouse with the Linoma tool was that it adds a bunch of file declaration keywords that are not necessary. The first thing I do after running a conversion with the tool is to manually clean it up.

Their conversion of D-specs does not apply any alignment to the resulting code - just inserts a single blank between the components. For me the result looks messy so again I apply some manual clean up after conversion. An example on the next chart.

Edited Version of Converted Code

I removed the superfluous file entries and applied my own alignment style to the data declarations.



Notes

Here you can see that I have removed the superfluous keywords from the file declaration. I have also added some alignment to the data declarations.

Right now my basic alignment rule is that in any given group (i.e. DS, PR, PI, series of standalone fields, etc.) I insert two spaces between the longest data name in the group and its data type/length definition and then align all other entries in the group to that. You can see the effect in the chart.

Note that I don't try and make all the entries in all the groups align - that is just too much work and I don't find it necessary - but visually I do like the entries in a specific group to align.

Choose your own style - but try to make sure that everyone in the shop uses the same (or very similar) style.

Learn more ...

Partner400

Free-Format RPG IV:

- Jon Paris' 5-part ***Geezer's Guide to Free-Form RPG*** series in IT Jungle
  - Part 1: https://www.itjungle.com/2014/02/12/fhg021214-story01/
  - Part 2: https://www.itjungle.com/2014/04/16/fhg041614-story01/
  - Part 3: https://www.itjungle.com/2014/05/28/fhg052814-story01/
  - Part 4: https://www.itjungle.com/2014/07/09/fhg070914-story01/
  - Part 5: https://www.itjungle.com/2014/08/13/fhg081314-story01/
- How to Bring Your RPG Programs Into the 21st Century
  - by Jim Martin - Get 3rd Edition or later for free format declarations
  - Published by MCPress (www.MC-store.com)

Linoma Software's RPG Toolbox

- Command interface + RDi plug-in
- www.linomasoftware.com/products/rpgtoolbox

Arcad Software's ARCAD Transformer RPG

- Command interface + RDi plug-in
- www.arcadsoftware.com/resource-items/arcad-transformer-rpg

Craig Rutledge's open source tools

- JCRCmds toolset contains some utilities related to Free-form
- More information at www.jcrcmds.com
- New open source project for RDi plug-in by Thomas Raddatz
  - https://sourceforge.net/projects/jcrcmds/

That's All Folks ...

Partner400

How to Code Logic in Free-form

Additional Materials

We don't have time to go into these charts but you may find them useful later

Replacing unsupported Op-codes

MOVE
ADDDUR & SUBDUR
Other Unsupported Op-Codes



Notes

We don't have time in this session to go into details about all the of the Free-form substitute methods, but we will take a look at some of the less obvious and/or most required ones.

Replacing MOVE - Like to Like

Partner400

If fields are the same length

- Replacement is EVAL

If both fields are character but of different lengths

- Replacement is (if blank padding is OK):
  - EVAL for MOVEL
  - EVALR for MOVE
- However, if padding is not desired
  - %Replace and/or %Subst will be needed
  - Or use a Data Structure

If both fields are numeric but of different length / decimals

- Replacement is EVAL but ...
  - An error will occur in the event of overflow
  - MOVE would have ignored it

Notes

Handling for character to character moves depends on the length of both the from and to fields.

Very often you will see code that moves blanks to a field and then moves another value in. Sometimes this is done even when the two fields are both character fields and the same length! At one point in time they might have been different lengths - but nobody removed the blank fill when the field length changed.

In fact the "P" extender (i.e. MOVE(P) or MOVEL(P)) to cause target fields to be blank filled has been around for some time and most blank fills could have been removed years ago!

It might not have occurred to you to use %SUBST to replace MOVE / MOVEL because most people don't realize that it can be used as part of the result field definition. So in a situation where MOVE is being used to replace the first three characters of a ten character filed, switching to EVAL and coding the result field as %SUBST(targetname: 1: 3) would have the same effect. It will also be much more obvious what is going on when yourself or others look at the program in future!

Replace MOVE - Character to Numeric

Use %DEC (expression { : precision : decimals })

- Expression can be character or numeric
- This capability also applies to %FLOAT, %INT, and %UNS
- As well as the Half adjust variants (%DECH, %INTH, and %UNSH)

Exception (Status 105) is signalled if character field not valid

- Use MONITOR to catch this error
  - See Rules and comments on "clean up" on Notes page

```
D CharField       S              12a   Inz(' -12345.678')
D NumField        S               9p 2
D Length          C                    %LEN(NumField)
D Decimals        C                    %DECPOS(NumField)

   NumField = %DecH( CharField : 9 : 2 );
   // This version "self adjusts" if NumField's definition changes
   NumField = %DecH( CharField: Length: Decimals);
```

Notes

Rules for format of character string:

Sign is optional. It can be '+' or '-', leading or trailing

The decimal point is optional

Blanks are allowed anywhere in the data.

For %DEC(H) both the Precision and Decimal Place parameters are compulsory

Floating point data (e.g. '2.5E4') is not allowed

In the example below, the assignment at <A> will work correctly. <B> though will cause an exception (Status code 105) because of the dollar ($) sign. IBM suggests a nice technique to handle this situation, as demonstrated at <C>. Simply put, the %XLATE BIF is applied to the character expression, substituting blanks for all unwanted characters (our example only replaces $ and , (comma). The resulting blanks will be ignored while processing the expression.

This deals with most common situations, but will not remove all sources of error. (e.g. there might be two decimal points in the field). One way to validate the character string is to use the subprocedure ChkNbr, which is part of the CGIDEV2 library available free from IBM. Go to www-922.ibm.com

If you do not wish to validate every character field before converting it, "Wrap" the operation in a MONITOR group (another wonderful V5R1 innovation) and catch the errors if and when they happen.

```
     D CharField1      S             14a   Inz('      1525.95-')
     D CharField2      S             14a   Inz('    $1,525.95-')
     D NumField        S              9  2
     D NumberEdits     C                   '$,'

<A>    NumField = %Dec(CharField1 : 9 : 2);
<B>    NumField = %Dec(CharField2 : 9 : 2);
<C>    NumField = %Dec( %Xlate('$,' : '  ' : CharField2) : 9 : 2);
```

Replace MOVE - Numeric to Character

Basic option - no decimal places

- First choice is to use the BIF %CHAR
  - But only if it's OK to suppress any leading zeros
- Alternative is to use %EDITC with the 'X' edit code
  - Retains leading zeros after conversion

If the numeric field contains decimal places

- The best bet is to use %EDITC with the 'X' edit code
  - And take the time to study exactly what you are trying to achieve

The BIFs will produce a character string

- Once the BIF has been selected and you know the length of the result apply the same rules as for Character to Character

```
D NumField        S              7P 0  Inz(12345)
D CharField       S              7A

  charField = %CHAR(numField);  // charField = '12345  '

  charField = %EDITC(numField: 'X');  // charField = '0012345'
```

Notes

It is worthwhile when replacing MOVEs to take the time determine exactly what was the intent of the original operation.

Note that %Char converts numeric fields to character, but it suppresses leading zeroes. If you want to retain leading zeroes in the character field, use %EDITC instead.

MOVE - Date to Numeric

Partner400

%DEC (date | time | timestamp { : format })

- Length of returned value is the number of digits in the Date
  - Or Time or Timestamp
- e.g. %DEC(date : *MDY) = length of 6
- Length of %DEC(Timestamp) is always 20

If format is not specified - the format of the date field is used

- In the example below it would have been *ISO and used 8 digits

```
D mmddyy          S               6S 0
 * DateFld uses the default (*ISO) format
D DateFld         S                D   Inz(D'2003-06-27')
 * Convert DateFld to numeric using MOVE
C     *MDY          Move      DateFld       mmddyy

  // Equivalent code Free-form code
   mmddyy = %Dec(DateFld: *MDY);
  // Result:  mmddyy = 062703
```

Notes

%DEC can be used to convert character strings and dates to numeric.

MOVE - Character/Numeric to Date

%DATE ({ expression { : date format } })

- Converts both character and numeric fields to date
- The second parameter specifies the format of the source field
  - Just as it did with MOVE
- If no parameters are passed then the system date is returned

Companion BIFs are %TIME and %TIMESTAMP

```
D CharDate1       S              6A   Inz('011549')
D CharDate2       S              6A   Inz('031954')
D ISODate         S               D   DatFmt(*ISO)
D USADate         S               D   DatFmt(*USA)

 * Loading a date field from a character field fixed form version
C     *MDY0         MOVE      CharDate1     ISODate
C     *MDY0         MOVE      CharDate2     USADate

  // Equivalent code in Free-form
   ISODate = %Date(CharDate1: *MDY0);
   USADate = %Date(CharDate2: *MDY0);
  // Expressions are also supported - e.g. Join 3 separate character fields
   USADate = %Date( Day + Month + Year): *DMY0);
```

Notes

The %DATE, %TIME and %TIMESTAMP BIFs are used to remove the requirement for a MOVE or MOVEL operation to convert data from character or numeric fields to date/time/timestamp fields.

Much like %CHAR will format date data into a character field, %DATE will do the reverse for character fields. The second (format) parameter specifies the format of the data being converted (i.e. the data represented by the first parameter). If the format is not specified, the default format for the program is used. If you recall, the default format for the program is the format specified with DATFMT (or TIMFMT for time fields) on the H spec or, if nothing is specified on the H spec, it will be *ISO.

Note that the format of the date (or time) returned will always be *ISO.

If you specify either *DATE or UDATE (to retrieve the job date) as the first parameter you should omit the format parameter altogether.

Note the difference between the job date and system date. *DATE or UDATE return the job date, %DATE() returns the system date (i.e. the current date). Also note that initializing a date field to *SYS using the D spec keyword only sets the INITIAL value for the field. So if a program runs over midnight, the initialized value in a date field defined with INZ(*SYS) will be different from that returned by %DATE().

Dates - Replacing SUBDUR

Calculating Durations

%DIFF(Date1 : Date2 : DurationType)

- Calculates durations between date, time or timestamp values
- Other duration calculations are performed by simple + and - operators
  - In conjunction with new duration Built-Ins

Duration types are as per ADDDUR / SUBDUR

- That is *MS, *S, *MN, *H, *D, *M, *Y
  - And the long-winded versions *MSECONDS, *SECONDS, *MINUTES, etc.

```
 * Is the loan due within the next 6 months ?
C     DueDate       SUBDUR    Today          MonthsToRun:*M
C                   IF        MonthsToRun < 6

   MonthsToRun = %Diff(DueDate : Today : *M );
   If MonthsToRun < 6;
   If %Diff(DueDate : Today : *M ) < 6; // Alternate coding
```

Notes

Hurray! Courtesy of the Free-form support, we finally have date duration support in expressions! This is a feature that RPG IV programmers have been wanting since V3R1. Most RPG IV programmers quickly became addicted to the extended factor 2 operation codes (e.g., EVAL, IF, etc.) and were dismayed to discover that in order to use the powerful date duration support built in to RPG IV from the beginning, they were forced to revert to the Factor 1, Factor 2, etc. format to use the powerful date duration operations.

The code examples here illustrate the ability to replace the ADDDUR and SUBDUR operation codes with expressions using either %DIFF or a simple arithmetic add (+) operation in combination with the duration BIF (%Years, in this example).

It also demonstrates that, because the new support is through BIFs, it can be used directly in expressions without the need for creating intermediate results.

On the next chart, we will show how to also replace the requirement for a MOVE operation code to get numeric data into a date or time field.

Now, all date data type operations can be done with free form operation codes.

Dates - Replacing ADDDUR/SUBDUR

Adding durations to a date, time or timestamp

- This function can now be performed by simple addition
- The durations are supplied via the appropriate BIFs
  - e.g. %MINUTES, %HOURS, %DAYS, %MONTHS, %YEARS, etc.
- Multiple durations can be handled in a single statement

Subtracting a duration works the same way

- i.e. It is achieved through the use of simple subtraction with BIFs

```
 * These original duration calculations
C     ContrDate      AddDur     Cyears:*Y      ExpDate
C     ExpDate        AddDur     CMonths:*M     ExpDate
C     ExpDate        AddDur     1:*D           expedite

C     ExpDate        SubDur     90:*D          WarnDate

 // Can be replaced by these Free-form calculations
  ExpDate = ContrDate + %Years(CYears) + %Months(CMonths) + %Days(1);

  WarnDate = ExpDate - %Days(90);
```

Notes

As on the previous chart, we are demonstrating here that since the new functionality is supplied by BIFs, they can be combined in a single expression - no need for the multiple and potentially error prone intermediate steps.

Combining The New Date Functions

Full power of the new BIFs appears when combining them

- In the example below it has significantly reduced the amount of code
  - The difference is actually greater than shown since in the new version there is no need to define the work fields Temp, Today and WorkDate
- %Date is used twice in the calculation
  - The first time to obtain the system date
  - The second to convert the numeric field DueDate to a "real" date

```
 * Determine if payment is more than 90 days overdue - prior to V5R1
C     *MDY0          MOVE      DueDate       WorkDate
C                    TIME                    Today

C     Today          SubDur    WorkDate      Temp:*D
C                    If        Temp > 90
C                    : ......  : ..............................
C                    EndIf

  // And the same calculation in Free-form
   If %Diff( %Date(): %Date(DueDate: *MDY0): *Days) > 90;
    ..............  ..........
   EndIf;
```

Returns the system date

Notes

In this example, you can see how much more powerful it is to be able to use date operations in expressions. In this small example alone, we did away with 3 temporary work fields. We think the new operation is also more obvious as to its purpose.

Note the use of %Date for 2 functions - first to retrieve the current date for the calculation and second to convert the DueDate field to a date form so it can be used in the duration calculation (via %DIFF).